1 Centrale TSI 99 – Math 2

```
> with(linalg):
> A:=matrix(5,5,(i,j)->if ((i+j)mod 5)=2 then 1 else 0 fi);
```

$$A := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \end{bmatrix}$$

```
> alpha:=convert(cos(2*Pi/5)+I*sin(2*Pi/5),radical);
```

$$\alpha := -\frac{1}{4} + \frac{1}{4}\sqrt{5} + \frac{1}{4} I \sqrt{2}\sqrt{5+\sqrt{5}}$$

```
> B:=matrix(5,5,(i,j)->convert(alpha^((i-1)*(j-1)),radical));
```

$$B := \begin{bmatrix} 1 & 1 & 1 & 1 & 1 \\ 1 & \%1 & \%1^2 & \%1^3 & \%1^4 \\ 1 & \%1^2 & \%1^4 & \%1^6 & \%1^8 \\ 1 & \%1^3 & \%1^6 & \%1^9 & \%1^{12} \\ 1 & \%1^4 & \%1^8 & \%1^{12} & \%1^{16} \end{bmatrix}$$

$$\%1 := -\frac{1}{4} + \frac{1}{4}\sqrt{5} + \frac{1}{4} I \sqrt{2}\sqrt{5+\sqrt{5}}$$

```
> SimplMat:=proc(M,n,p)
> local i,j,N;
> N:=matrix(n,p,(i,j)->simplify(evalm(M)[i,j]));
> evalm(N)
> end;
```

SimplMat := **proc**(*M*, *n*, *p*)
local *i*, *j*, *N*;
N := matrix(*n*, *p*, (*i*, *j*) → simplify(evalm($M_{i,j}$)) ; evalm(*N*)
end proc

```
> B2:=SimplMat(B*B,5,5);
```

$$B2 := \begin{bmatrix} 5 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 5 \\ 0 & 0 & 0 & 5 & 0 \\ 0 & 0 & 5 & 0 & 0 \\ 0 & 5 & 0 & 0 & 0 \end{bmatrix}$$

```
> P:=matrix([[1,0,0,0,0],[0,1,0,1,0],[0,0,1,0,1],[0,0,1,0,-1],[0,1,0,-1
> ,0]]);
```

$$P := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 1 & 0 \\ 0 & 0 & 1 & 0 & 1 \\ 0 & 0 & 1 & 0 & -1 \\ 0 & 1 & 0 & -1 & 0 \end{bmatrix}$$

```
> A1:=evalm(P^(-1)&*A&*P);
```

$$A1 := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & 0 & -1 \end{bmatrix}$$

```
> B1:=SimplMat(P^(-1)&*B&*P,5,5);
```

$$B1 := \begin{bmatrix} 1 & 2 & 2 & 0 & 0 \\ 1 & -\frac{1}{2}+\frac{1}{2}\sqrt{5} & -\frac{1}{2}-\frac{1}{2}\sqrt{5} & 0 & 0 \\ 1 & -\frac{1}{2}-\frac{1}{2}\sqrt{5} & -\frac{1}{2}+\frac{1}{2}\sqrt{5} & 0 & 0 \\ 0 & 0 & 0 & \frac{1}{2} I \sqrt{2}\,\%1 & \frac{1}{4} I \sqrt{5}\,\sqrt{2}\,\%1 - \frac{1}{4} I \sqrt{2}\,\%1 \\ 0 & 0 & 0 & \frac{1}{4} I \sqrt{5}\,\sqrt{2}\,\%1 - \frac{1}{4} I \sqrt{2}\,\%1 & \frac{-1}{2} I \sqrt{2}\,\%1 \end{bmatrix}$$

$$\%1 := \sqrt{5+\sqrt{5}}$$

```
> ExtraitMat:=proc(M,q,r,n0,p0)
> local i,j,N;
> N:=matrix(q,r,(i,j)->M[i+n0-1,j+p0-1]);
> evalm(N)
> end;
```

$$ExtraitMat := \mathbf{proc}(M, q, r, n0, p0)$$
$$\mathbf{local}\ i, j, N;$$
$$N := \mathrm{matrix}(q, r, (i, j) \to M_{i+n0-1,\, j+p0-1}); \mathrm{evalm}(N)$$
$$\mathbf{end\ proc}$$

```
> X:=ExtraitMat(B1,3,3,1,1);
```

$$X := \begin{bmatrix} 1 & 2 & 2 \\ 1 & -\frac{1}{2}+\frac{1}{2}\sqrt{5} & -\frac{1}{2}-\frac{1}{2}\sqrt{5} \\ 1 & -\frac{1}{2}-\frac{1}{2}\sqrt{5} & -\frac{1}{2}+\frac{1}{2}\sqrt{5} \end{bmatrix}$$

```
> Y:=ExtraitMat(B1,2,2,4,4);
```

$$Y := \begin{bmatrix} \frac{1}{2} I \sqrt{2}\,\%1 & \frac{1}{4} I \sqrt{5}\,\sqrt{2}\,\%1 - \frac{1}{4} I \sqrt{2}\,\%1 \\ \frac{1}{4} I \sqrt{5}\,\sqrt{2}\,\%1 - \frac{1}{4} I \sqrt{2}\,\%1 & \frac{-1}{2} I \sqrt{2}\,\%1 \end{bmatrix}$$

$$\%1 := \sqrt{5+\sqrt{5}}$$

```
> eigenvects(X);
```

$$[-\sqrt{5},\ 1,\ \{\left[1, -\frac{1}{4}\sqrt{5}-\frac{1}{4}, -\frac{1}{4}\sqrt{5}-\frac{1}{4}\right]\}],\ [\sqrt{5},\ 2,\ \{[0, 1, -1], \left[1, 0, -\frac{1}{2}+\frac{1}{2}\sqrt{5}\right]\}]$$

```
> Q:=matrix([[1,1,0],[-1/4*sqrt(5)-1/4,0,1],[-1/4*sqrt(5)-1/4,1/2*sqrt(
> 5)-1/2,-1]]);
```

$$Q := \begin{bmatrix} 1 & 1 & 0 \\ -\frac{1}{4}\sqrt{5}-\frac{1}{4} & 0 & 1 \\ -\frac{1}{4}\sqrt{5}-\frac{1}{4} & -\frac{1}{2}+\frac{1}{2}\sqrt{5} & -1 \end{bmatrix}$$

```
> X1:=SimplMat(Q^(-1)&*X&*Q,3,3);
```

$$X1 := \begin{bmatrix} -\sqrt{5} & 0 & 0 \\ 0 & \sqrt{5} & 0 \\ 0 & 0 & \sqrt{5} \end{bmatrix}$$

```
> eigenvects(Y);
```

$$[-I\sqrt{5}, 1, \{\left[\frac{1}{2}+\frac{1}{2}\sqrt{5}-\frac{1}{2}\sqrt{2}\sqrt{5+\sqrt{5}}, 1\right]\}], [I\sqrt{5}, 1, \{\left[\frac{1}{2}+\frac{1}{2}\sqrt{5}+\frac{1}{2}\sqrt{2}\sqrt{5+\sqrt{5}}, 1\right]\}]$$

```
> R:=matrix([[-1/2*sqrt(2)*sqrt(5+sqrt(5))+1/2*sqrt(5)+1/2,1/2*sqrt(2)*
> sqrt(5+sqrt(5))+1/2*sqrt(5)+1/2],[1,1]]);
```

$$R := \begin{bmatrix} \frac{1}{2}+\frac{1}{2}\sqrt{5}-\frac{1}{2}\sqrt{2}\sqrt{5+\sqrt{5}} & \frac{1}{2}+\frac{1}{2}\sqrt{5}+\frac{1}{2}\sqrt{2}\sqrt{5+\sqrt{5}} \\ 1 & 1 \end{bmatrix}$$

```
> Y1:=SimplMat(R^(-1)&*Y&*R,2,2);
```

$$Y1 := \begin{bmatrix} -I\sqrt{5} & 0 \\ 0 & I\sqrt{5} \end{bmatrix}$$

```
> P1:=matrix(5,5,(i,j)->if (i<=3 and j<=3) then Q[i,j] elif (i>=4 and
> j<=3) then 0 elif (i<=3 and j>=4) then 0
> else R[i-3,j-3] fi);
```

$$P1 := \begin{bmatrix} 1 & 1 & 0 & 0 & 0 \\ -\frac{1}{4}\sqrt{5}-\frac{1}{4} & 0 & 1 & 0 & 0 \\ -\frac{1}{4}\sqrt{5}-\frac{1}{4} & -\frac{1}{2}+\frac{1}{2}\sqrt{5} & -1 & 0 & 0 \\ 0 & 0 & 0 & \frac{1}{2}+\frac{1}{2}\sqrt{5}-\frac{1}{2}\sqrt{2}\sqrt{5+\sqrt{5}} & \frac{1}{2}+\frac{1}{2}\sqrt{5}+\frac{1}{2}\sqrt{2}\sqrt{5+\sqrt{5}} \\ 0 & 0 & 0 & 1 & 1 \end{bmatrix}$$

```
> SimplMat(P1^(-1)&*B1&*P1,5,5);
```

$$\begin{bmatrix} -\sqrt{5} & 0 & 0 & 0 & 0 \\ 0 & \sqrt{5} & 0 & 0 & 0 \\ 0 & 0 & \sqrt{5} & 0 & 0 \\ 0 & 0 & 0 & -I\sqrt{5} & 0 \\ 0 & 0 & 0 & 0 & I\sqrt{5} \end{bmatrix}$$

```
> P2:=SimplMat(P&*P1,5,5);
```

$$P2 := \begin{bmatrix} 1 & 1 & 0 & 0 & 0 \\ \%2 & 0 & 1 & \frac{1}{2}+\frac{1}{2}\sqrt{5}-\frac{1}{2}\sqrt{2}\,\%1 & \frac{1}{2}+\frac{1}{2}\sqrt{5}+\frac{1}{2}\sqrt{2}\,\%1 \\ \%2 & -\frac{1}{2}+\frac{1}{2}\sqrt{5} & -1 & 1 & 1 \\ \%2 & -\frac{1}{2}+\frac{1}{2}\sqrt{5} & -1 & -1 & -1 \\ \%2 & 0 & 1 & -\frac{1}{2}-\frac{1}{2}\sqrt{5}+\frac{1}{2}\sqrt{2}\,\%1 & -\frac{1}{2}-\frac{1}{2}\sqrt{5}-\frac{1}{2}\sqrt{2}\,\%1 \end{bmatrix}$$

$$\%1 := \sqrt{5+\sqrt{5}}$$

$$\%2 := -\frac{1}{4}\sqrt{5}-\frac{1}{4}$$

2 CCP TSI 97 – Math 2

```
> with(linalg):
> A:=matrix([[4,0,1],[0,4,-1],[-2,1,8]]);
```

$$A := \begin{bmatrix} 4 & 0 & 1 \\ 0 & 4 & -1 \\ -2 & 1 & 8 \end{bmatrix}$$

```
> Y:=vector([6,-6,13]);
```

$$Y := [6, -6, 13]$$

```
> ResolApproxSyst:=proc(A,Y,e)
> local B,Z,X1,X2,D,E,mu;
> B:=evalm(transpose(A)&*A);
> Z:=evalm(transpose(A)&*Y);
> X1:=vector([0,0,0]);
> D:=evalm(B&*X1-Z);
> E:=evalm(B&*D);
> mu:=-evalf(dotprod(D,E)/norm(E,2)**2);
> X2:=evalm(X1+mu*D);
> while evalf(norm(evalm(X2-X1),2))>evalf(e) do
> X1:=evalm(X2);
> D:=evalm(B&*X1-Z);
> E:=evalm(B&*D);
> mu:=-evalf(dotprod(D,E)/(norm(E,2)**2));
> X2:=evalm(X1+mu*D);
> od;
> evalm(X2)
> end;
```

```
ResolApproxSyst := proc(A, Y, e)
local B, Z, X1, X2, D, E, mu;
  B := evalm(`&*`(transpose(A), A));
  Z := evalm(`&*`(transpose(A), Y));
  X1 := vector([0, 0, 0]);
  D := evalm(`&*`(B, X1) - Z);
  E := evalm(`&*`(B, D));
  mu := -evalf(dotprod(D, E)/norm(E, 2)^2);
  X2 := evalm(X1 + mu*D);
  while evalf(e) < evalf(norm(evalm(X2 - X1), 2)) do
    X1 := evalm(X2);
    D := evalm(`&*`(B, X1) - Z);
    E := evalm(`&*`(B, D));
    mu := -evalf(dotprod(D, E)/norm(E, 2)^2);
    X2 := evalm(X1 + mu*D)
  end do;
  evalm(X2)
end proc
```

```
> ResolApproxSyst(A,Y,0.00000001);
```

$$[.9999999998, -1.000000001, 2.000000001]$$